Solution to Problem 1201.

Problem 1201 : A random walk on the 2-dimensional integer lattice begins at the origin. At each step, the walker moves one unit either left, right, or up, each with probability 1/3. (No downward steps ever.) A walk is a success if it reaches the point (1, 1). What is the probability of success? More generally, what is $p(x, y)$, the probability of reaching (x, y)?

I received solutions to various aspects of the problem from John Sullivan, Bruce Torrence, Rob Pratt, John Snyder, Lee Gao, Joseph DeVincentis, Sam Vandervelde, Wei Chen, and Barry Cox. Torrence and Snyder worked hard on various aspects, and the notes below are largely based on their work, with several enhancements by me. I forgot that chapter 6 of our book, "The SIAM 100-Digit Challenge", studies a related problem on this sort of random walk, and the methods there are likely applicable.

Notation: ϕ = golden ratio = $\frac{1}{2}\left(1+\sqrt{5}\right)$; $\lambda = \frac{1}{2}\left(3-\sqrt{5}\right) = \phi^{-2}$;

F_n is the nth Fibonacci number;
$p(x, y)$ is the probability of eventual arrival at (x, y).
All computer code is *Mathematica*.

Problem A: $p(n, 0) = \lambda^{|n|} = \phi^{-2|n|} = F_{2n}\lambda - F_{2n-2}$

0. $p(0, 0) = 1$
1. If $y < 0$, then $p(x, y) = 0$.
2. $p(x, y) = p(-x, y)$
3. $p(x, y) = \frac{1}{3}(p(x-1, y) + p(x, y-1) + p(x+1, y)$
4. $p(2, 0) = p(1, 0)^2$
5. $p(n, 0) = p(1, 0)^{|n|}$
6. $p(1, 0) = \lambda = \phi^{-2}$, where ϕ is the golden ratio
7. $p(n, 0) = \lambda^{|n|} = \phi^{-2|n|}$
8. $p(n, 0) = -F_{2n-2} + F_{2n}\lambda$

0–2: easy.

3. For $0 \le j$, let $q_{i,j}(a, b)$ be the probability of eventual arrival at (a, b) given that the walker starts at (i, j). So $p(a, b) = q_{0,0}(a, b)$. We have two simple identities:

(i) $q_{i,j}(a, b) = p(a-i, b-j)$

(ii) $q_{0,0}(a, b) = \frac{1}{3}(q_{1,0}(a, b) + q_{0,1}(a, b) + q_{-1,0}(a, b))$

Translation proves (i): Getting from (i, j) to (a, b) is the same as getting from (0, 0) to $(a-i, b-j)$. The proof of (ii) is by considering the first step the walker takes. If he is at (0, 0), after one step he has a $\frac{1}{3}$ chance of being at each of (1, 0), (0, 1), or (−1, 0). The proof of the recurrence in (3) is now simple. By (i) and (ii),

$$p(a, b) = q_{0,0}(a, b) = \frac{1}{3}(q_{1,0}(a, b) + q_{0,1}(a, b) + q_{-1,0}(a, b)) = \frac{1}{3}(p(a-1, b) + p(a, b-1) + p(a-1, b)).$$

4. The only way of reaching (2, 0) is by first reaching (1, 0). Hence the problem is the same as the original problem times the probability of going from (1, 0) to (2, 0); but the latter is, by translation, the same as $p(1, 0)$.

5. Same as (4).

6. By (0), (1), (3), (4): $p(1,0) = \frac{1}{3}(1 + 0 + p(2,0)) = \frac{1}{3}\left(1 + 0 + p(1,0)^2\right)$. The the quadratic is solved to give the result.

7. By (5) and (6).

8. Fairly simple algebra using (7). Here is how to do it using the Binet formula expressing Fibonacci number in terms of ϕ

```
Fib[n_] := (GoldenRatio^n - GoldenRatio^-n)/Sqrt[5]

Simplify[RootReduce[GoldenRatio^(-2 n) == (-Fib[2 n - 2] + Fib[2 n]/GoldenRatio^2)],
 n > 0 && n ∈ Integers]
True
```

Problem B: $p(0, 1) = \frac{3}{5}$

To get from (0, 0) to (0, 1) one must wander along the x-axis, move up at some $(k, 0)$ (assume that $k \geq 0$), and then wander horizontally back to the y-axis. This last step will occur with probability $p(k, 0)$ since it is the same as problem A. The up step occurs with probability $\frac{1}{3}$. The initial walk can involve $2n + k$ steps, where exactly n of the steps are "Left". Therefore, with k fixed, this sum gives the $p(0, 1)$: $\frac{1}{3}\lambda^k \sum_{n=0}^{\infty}\binom{2n+k}{n}3^{-(2n+k)}$. This sum is just $\frac{1}{\sqrt{5}}\lambda^{2k}$.

```
Simplify[FunctionExpand[1/3 λ^k Sum[Binomial[2 n + k, n]/3^(2 n + k), {n, 0, ∞}]]] /. 2/(3 + Sqrt[5]) → λ
```

$$\frac{\lambda^{2k}}{\sqrt{5}}$$

Now we sum this as k, the move-up point, varies from 1 to ∞, double the result to account for k negative, and add $\frac{1}{\sqrt{5}}$, which is the sum for $k = 0$. The result is $\frac{3}{5}$.

```
λRule = λ → 1/GoldenRatio^2;
FullSimplify[
 Sum[Binomial[2 n, n]/3^(2 n + 1), {n, 0, ∞}] + 2 Sum[FunctionExpand[1/3 λ^k Sum[Binomial[2 n + k, n]/3^(2 n + k), {n, 0, ∞}]], {k, 1, ∞}] /. λRule]
```

$$\frac{3}{5}$$

Aside on sums. Expressing the infinite sums in simpler form uses the standard technique of generating functions. To present just one example, consider the $k = 0$ case and view 3^{2n+1} as being $3 \cdot 9^n$, and then consider $\frac{1}{9}$ as being x. Then this formula is key: $f(x) = \sum_{n=0}^{\infty}\binom{2n}{n}x^n = \frac{1}{\sqrt{1-4x}}$. Let c_n be the binomial coefficient. Then $(n+1)\, c_n = (4\, n+2)\, c_{n+1}$. Differentiation of the series and applying this identity yields the differential equation $f'(x) = 4\, x\, f'(x) + 2\, f(x)$. This is easily solved by standard techniques.

```
f[x] /. DSolve[f'[x] == 4 x f'[x] + 2 f[x] && f[0] == 1 , f[x], x] [[1]]
```

$$\frac{1}{\sqrt{1-4\,x}}$$

And then the $k = 0$ value of $\frac{1}{\sqrt{5}}$ arises by letting $x = \frac{1}{9}$ and multiplying by the extra $\frac{1}{3}$.

$$\frac{1}{3}\,\frac{1}{\sqrt{1-4\left(\frac{1}{9}\right)}}$$

$$\frac{1}{\sqrt{5}}$$

Problem C: $p\,(1,\,1) = \frac{2}{5}$

This follows from (0), (2), (3), and Problem B. The equation to be solved is $\frac{1}{3}\,(2\,x+1) = \frac{3}{5}$. So $x = \frac{2}{5}$.

Problem D:
$p(n,\,1) =$

$$\lambda^{|n|}\left(\frac{1}{\sqrt{5}}\,|n| + \frac{3}{5}\right) = \frac{1}{5}\,(2\,n\,F_{2\,n} - 3\,(n+1)\,F_{2\,n-2}) - \lambda\,(3\,(n-1)\,F_{2\,n} - 2\,n\,F_{2\,n-2})$$

The technique of a double sum using binomial coefficients works for $p\,(n,\,1)$.

The following code gets it, though it is not very pretty in this form.

```
FullSimplify[Total[{Sum[FunctionExpand[1/3 λ^(k-j) Sum[Binomial[2 n + k, n]/3^(2 n+k), {n, 0, ∞}]],
    {k, j, ∞}, Assumptions → j ≥ 0 && j ∈ Integers] /. λRule,
   Sum[FunctionExpand[1/3 λ^(j-k) Sum[Binomial[2 n + k, n]/3^(2 n+k), {n, 0, ∞}] /. λRule],
    {k, 0, j - 1}, Assumptions → j > 0 && j ∈ Integers] /. λRule,
   Sum[FunctionExpand[1/3 λ^(j+k) Sum[Binomial[2 n + k, n]/3^(2 n+k), {n, 0, ∞}]], {k, 1, ∞},
    Assumptions → j ≥ 0 && j ∈ Integers] /. λRule}], j ∈ Integers && j > 0]
```

$$\frac{\left(1+\sqrt{5}\right)^{-2\,j}\left(4^{j}\left(3+\sqrt{5}\right)+\left(1+\sqrt{5}\right)^{2\,j}\left(\frac{2}{3+\sqrt{5}}\right)^{j}\left(18+8\,\sqrt{5}+\left(15+7\,\sqrt{5}\right)\,j\right)\right)}{5\left(7+3\,\sqrt{5}\right)}$$

Bruce Torrence and Lee Gao each used a slightly different approach and got the following simpler formula (which equals the preceding).

$$p(n,\,1) = \frac{\lambda^{|n|}}{\sqrt{5}}\left(|\,n\,| + \frac{1+\lambda^2}{1-\lambda^2}\right) = \frac{\lambda^{|n|}}{\sqrt{5}}\left(|\,n\,| + \frac{3}{\sqrt{5}}\right) = \lambda^{|n|}\left(\frac{1}{\sqrt{5}}\,|\,n\,| + \frac{3}{5}\right)$$

Some algebra on this yields an expression in terms of Fibonacci numbers.

$$p(n,\,1) = \frac{1}{5}\left(2\,n\,F_{2\,n} - 3\,(n+1)\,F_{2\,n-2}\right) - \lambda\left(3\,(n-1)\,F_{2\,n} - 2\,n\,F_{2\,n-2}\right)$$

Problem E: A general formula: $p(x,\,y) = \sqrt{5}\;3^{-(1+x+y)}\binom{x+y}{x}\;{}_2F_1\left(\frac{1}{2}\,(1+x+y),\,\frac{1}{2}\,(2+x+y);\,1+x;\,\frac{4}{9}\right)$

The following was found by John Snyder, using lattice Green functions and several other ideas. His complete notes are posted at <http://mathforum.org/wagon/current_solutions/s1201_2.pdf>. The notation ${}_2F_1$ refers to the Gaussian (or "ordinary") hypergeometric function.

$$p(x,\,y) = \frac{\sqrt{5}}{3^{1+x+y}}\binom{x+y}{x}\;{}_2F_1\left(\frac{1}{2}\,(1+x+y),\,\frac{1}{2}\,(2+x+y);\,1+x;\,\frac{4}{9}\right)$$

To prove this is correct is rather simple: we need only check that it satisfies the basic recursion (3) and some base cases. Because the definition of the hypergeometric function ${}_2F_1$ involves multinomial coefficients, it is presumably not too hard to verify what follows by hand. The definition below assumes that x is nonnegative.

```
P[x_, y_] :=
  3^-(1+x+y) √5 Binomial[x + y, x] Hypergeometric2F1[1/2 (1 + x + y), 1/2 (2 + x + y), 1 + x, 4/9];
```

```
FullSimplify[FunctionExpand[(P[x + 1, y] + P[x - 1, y] + P[x, y - 1]) - 3 P[x, y]],
 y ≥ 0 && x ≥ 0 && (x | y) ∈ Integers]
```

0

And the base row:

```
Simplify[FunctionExpand[P[n, 0]], n ≥ 0] /. 1/2 (3 - √5) → λ
```

λ^n

Here then is a table of values of $5^y\,p(x,\,y)$.

```
Simplify[Grid[aa = Append[
      Table[Simplify[P[Abs[x], y]] 5^y, {y, 7, 0, -1}, {x, 0, 7}], Range[0, 7]];
    Table[Prepend[aa[[i]], ToString[8 - i]], {i, 9}] /. "-1" → "",
    Dividers → {AbsoluteThickness /@ {1, 1, 0, 0, 0, 0, 0, 0, 0, 1},
      AbsoluteThickness /@ {1, 0, 0, 0, 0, 0, 0, 0, 1, 1}}, BaseStyle →
   {FontFamily → "Ariel", 10}] /. 1/2 (3 - √5) → λ /. √5 → 3 - 2 λ] // Simplify
```

7	18483	17752	15768	13032	10098	7392	5148	3432
6	3989	3801	3304	2646	1974	1386	924	$-5379+15625\lambda$
5	873	822	693	532	378	252	$1353-3125\lambda$	$21582-56250\lambda$
4	195	180	145	105	70	$-195+625\lambda$	$-3555+9375\lambda$	$-31020+81250\lambda$
3	45	40	30	20	$60-125\lambda$	$580-1500\lambda$	$4110-10750\lambda$	$22920-60000\lambda$
2	11	9	6	$-6+25\lambda$	$-84+225\lambda$	$-486+1275\lambda$	$-2234+5850\lambda$	$-9081+23775\lambda$
1	3	2	$3-5\lambda$	$12-30\lambda$	$48-125\lambda$	$172-450\lambda$	$573-1500\lambda$	$1822-4770\lambda$
0	1	λ	$-1+3\lambda$	$-3+8\lambda$	$-8+21\lambda$	$-21+55\lambda$	$-55+144\lambda$	$-144+377\lambda$
	0	1	2	3	4	5	6	7

A consequence of all this work is the following:

$p(n, 0) = -F_{2n-2} + F_{2n}\lambda$, where F_n is the nth Fibonacci number.

$p(0, n) = \frac{1}{5^n} \sum_{k=0}^{n} \binom{n}{k} \binom{2k}{k}$

$p(n, n) = \frac{1}{5^n} \binom{2n}{n}$

And I'll repeat this one:

$5\,p(n, 1) = (2n F_{2n} - 3(n+1) F_{2n-2}) - \lambda(3(n-1) F_{2n} - 2n F_{2n-2})$

It seems that, when y is fixed, $p(n, y)$ has the form $a + b\lambda$ where a and b are expressions involving the Fibonacci numbers, but I have not worked out any more details. Except..... Some additional work on the next level by Bruce and me leads to the following formula for $5^2\,p(n, 2)$ (the negative Fibonacci numbers here are $F_{-2} = -1$ and $F_{-1} = 1$.

$$5^2\,p(n, 2) = \left(-\left(\frac{5}{2}n^2 + \frac{27}{2}n + 11\right) F_{2n-2} + 9n F_{2n}\right) + \lambda\left(\left(\frac{5}{2}n^2 - \frac{27}{2}n + 11\right) F_{2n} + 9n F_{2n-2}\right)$$

Here is a check:

```
Table[1/2 (-5 n^2 - 27 n - 22) F[2 n - 2] + 9 n F[2 n] +
   λ (1/2 (5 n^2 - 27 n + 22) F[2 n] + 9 n F[2 n - 2]) /. F → Fibonacci, {n, 0, 7}]
{11, 9, 6, -6 + 25 λ, -84 + 225 λ, -486 + 1275 λ, -2234 + 5850 λ, -9081 + 23 775 λ}
```

And then guessing at the form for the next row and using the data to find the coefficients yields this for $5^3\,p(n, 3)$. Let $Q_n = \frac{5}{3}(n^3 + 23n)$ and

$P_n = \frac{5n^3}{2} + 15n^2 + \frac{115n}{2} + 45$:

$$5^3\,p(n, 3) = (Q_n F_{2n} - P_n F_{2n-2}) + \lambda(P_{-n} F_{2n} + Q_n F_{2n-2})$$

Note how the 45 appears here (and the 11 in the one above); those are the numbers that would start the horizontal induction. It appears that the general form for $p(n, j)$ is the following, where P_n and Q_n are polynomials in n of degree j.

$$\begin{pmatrix} P_n & Q_n \\ Q_n\lambda & P_{-n}\lambda \end{pmatrix} \begin{pmatrix} F_{2n-2} \\ F_{2n} \end{pmatrix}$$

I have now automated the process and can solve the linear equations needed to generate these various polynomia-plus-Fibonacci formulas. It seems clear that such a formula exists for each row.